

0

3

```

0          ( -----overlay prefix for ENGINEER----- )
1
2          VOCABULARY ENGINEER-VOC IMMEDIATE
3          98 OPEN-OVERLAY
4          ENGINEER-VOC DEFINITIONS
5
6          2000 TRANS-ALLOT NEWT-DP
7
8
9          16 28 1 IFIELD: ^ENG      17 26 3 IFIELD: ENGINEER
10         20 17 2 IFIELD: %ARMOR
11
12         HEAD: .CTY/N T: ' ov.CTY/N MODULE T;
13
14
15

```

1

4

```

0          ( Crew fcns: Engineer                                AMK 7/02/85 )
1
2          HEAD: .DAMAGE-GRAPHICS ( -- )
3          T: DK-BLUE !COLOR
4          154 180 113 OVER LLINE 154 130 113 OVER LLINE
5          113 131 OVER 179 LLINE 154 131 OVER 179 LLINE T;
6
7          HEAD: NXT-DMG ( -- ) T: -7 vBLT +! 85 xBLT ! T;
8
9          HEAD: .DAMAGE-TEXT ( -- )
10         T: >2FONT WHITE !COLOR 103 194 POS. ." DAMAGE"
11         >1FONT DK-BLUE !COLOR 85 178 POS. ." HULL"
12         NXT-DMG ." ENGINES" NXT-DMG ." SENSORS" NXT-DMG ." COMM."
13         NXT-DMG ." SHIELD" NXT-DMG ." MISSILE" NXT-DMG ." LASER" T;
14
15

```

2

5

```

0          ( Crew fcns: Engineer                                AMK 10/02/85 )
1
2          HEAD: .DAMAGE-BAR ( n -- )
3          T: DUP 2* %ARMOR + >R I 1+ C@ I %ARMOR = OR
4          IF CURSORSPACE aBLT ! XORMODE OFF 5 1BLT ! I C@ 2* 5 / wBLT !
5          114 178 ROT 7 * - POS. DK-GREEN !COLOR wBLT @
6          IF BLT THEN
7          114 wBLT @ + xBLT ! 40 wBLT @ - wBLT ! RED !COLOR wBLT @
8          IF BLT THEN ELSE DROP THEN R> DROP T;
9
10         : DBARS ( -- )
11         *SHIP >C+S 7 0 DO I .DAMAGE-BAR LOOP ICLOSE ;
12
13         : (/DAMAGE) ( -- )
14         BLACK ERASE-AUXILLARY .DAMAGE-GRAPHICS
15         .DAMAGE-TEXT DBARS ;

```

6

```

0 ( Crew fcns: Engineer
1
2 HEAD: SCAN-DAMAGE-STATS ( n1 -- [n2] f )
3 T: >R *SHIP >C+S 0
4 7 0 DO %ARMOR I 2* + 1+ C@ J AND
5   IF 1+ I SWAP LEAVE THEN
6 LOOP R> DROP ICLOSE T;
7
8 HEAD: .SHIP-VITALS ( n -- )
9 T: >R I 0= IF ." DESTROYED" ELSE
10   I 1 20 WITHIN IF ." CRITICALLY" ELSE
11   I 20 40 WITHIN IF ." HEAVILY" ELSE
12   I 40 70 WITHIN IF ." MODERATELY" ELSE
13   I 70 90 WITHIN IF ." SLIGHTLY" ELSE
14   I 90 100 WITHIN IF ." MINIMALLY" ELSE ." NOT"
15 THEN THEN THEN THEN THEN ." DAMAGED" THEN R> DROP T;

```

7

```

0 ( Crew fcns: Engineer
1
2 HEAD: .S10 ( n -- ) T: ." HULL" T;
3 HEAD: .S11 ( n -- ) T: ." ENGINES" T;
4 HEAD: .S12 ( n -- ) T: ." SENSORS" T;
5 HEAD: .S13 ( n -- ) T: ." COMMUNICATIONS" T;
6 HEAD: .S14 ( n -- ) T: ." SHIELDS" T;
7 HEAD: .S15 ( n -- ) T: ." MISSILE LAUNCHER" T;
8 HEAD: .S16 ( n -- ) T: ." LASER CANNON" T;
9
10 CASE .SHIP-ITEM
11 0 IS .S10 1 IS .S11 2 IS .S12
12 3 IS .S13 4 IS .S14 5 IS .S15 6 IS .S16
13 OTHERS NOP
14
15

```

8

```

0 ( Crew fcns: Engineer
1
2 HEAD: .DL1 ( n -- f )
3 T: NULL CTPOS. DUP .SHIP-ITEM ." : "
4 2* %ARMOR + >R I 1+ C@ I %ARMOR = OR
5 IF I C@ .SHIP-VITALS I C@ 0> ELSE ." NONE" 0 THEN R> DROP T;
6
7 HEAD: .DL2 ( n -- )
8 T: CTCR ." EFFICIENCY: " 2* %ARMOR + C@ 0 .R ." %" T;
9
10 HEAD: .DL3 ( n -- )
11 T: ?DUP IF CTCR ." CURRENT FUNCTIONALITY: " 2* %ARMOR + 1+ C@
12 128 AND IF ." IN" THEN ." OPERATIVE" THEN T;
13
14
15

```

9

```

AWK 10/23/84 ) ( Crew fcns: Engineer AWK 10/23/84 )
HEAD: GREG'S-FORMULA ( %eff -- n )
T: *ASSIGN-CREW >C+S ENGINEER 1.5@ >C+S
250 ^ENG C@ - 100 ROT - * 1+ 7195 21849 */ 5 + CDROP ICLOSE T;

HEAD: .TIME-UNITS ( n -- )
T: DUP 60 < IF 5 /MOD SWAP 2 > + DUP . ." MINUTE" ELSE
  DUP 1440 < IF 60 /MOD SWAP 30 > + DUP . ." HOUR" ELSE
  1440 /MOD SWAP 720 > + DUP . ." DAY"
  THEN THEN 1 > IF 83 EMIT THEN ." ." T;

HEAD: .DL4 ( n -- )
T: 2* %ARMOR + C@ DUP 100 <
  IF CTCR ." ESTIMATED TIME FOR COMPLETE REPAIRS: "
  CTCR GREG'S-FORMULA .TIME-UNITS ELSE DROP THEN T;

```

10

```

AWK 10/23/84 ) ( Crew fcns: Engineer AWK 10/23/84 )
HEAD: .DL5 ( n -- )
T: 2* %ARMOR + 1+ C@ 64 AND
  IF CTCR ." CURRENTLY UNDER REPAIR." THEN T;

HEAD: .DAMAGE ( n -- )
T: *SHIP >C+S
  >R CTINIT CTERASE I .DL1
  IF I .DL2 I .DL3 I .DL4 I .DL5 THEN
  R> DROP ICLOSE T;

HEAD: .?DESTROYED ( n -- f )
T: DUP 2* %ARMOR + C@ 0=
  IF CTERASE .SHIP-ITEM ." : DAMAGED BEYOND REPAIR." 1
  ELSE DROP 0 THEN T;

```

11

```

AWK 10/23/84 ) ( Crew fcns: Engineer AWK 10/14/85 )
HEAD: .REPAIRING ( n -- )
T: CTERASE .SHIP-ITEM ." : "
  CTCR ." ALREADY UNDERGOING REPAIRS." T;

HEAD: .?NEW-REPAIR ( n -- f )
T: CTERASE ." CEASE REPAIRS ON " .SHIP-ITEM
  CTCR ." AND BEGIN REPAIRS ON " OCRS @ .SHIP-ITEM ." ?"
  CTCR .CTY/N Y/N FQUIT OFF CTERASE T;

HEAD: .REPAIR ( n -- )
T: ." CEASING REPAIRS ON " .SHIP-ITEM ." ." T;

HEAD: .REPAIR ( n -- )
T: DUP .REPAIR 2* %ARMOR + 1+ DUP C@ 191 AND SWAP C! T;

```

12

```

0 ( Crew fcns: Engineer                               AWK 10/23/84 )
1
2 HEAD: .CONT-REPAIR ( n -- )
3 T: ." CONTINUING REPAIRS ON " .SHIP-ITEM ." ." T;
4
5 HEAD: >REPAIR ( n -- )
6 T: ." BEGINNING REPAIRS ON " .SHIP-ITEM ." ." ?REPAIR ON T;
7
8 HEAD: >REPAIR ( n -- )
9 T: DUP .>REPAIR *SHIP >C+S 2* %ARMOR + 1+ DUP
10 C@ 64 OR SWAP C! ICLOSE T;
11
12 HEAD: .REPAIR-CRS ( n -- )
13 T: >R @CRS I OCRS ! I 84 179 ROT 7 * - POS. CURSORSPACE aBLT !
14 29 wBLT ! 7 1BLT ! LT-BLUE !COLOR XORMODE ON
15 BLT !CRS R> OCRS ! T;

```

13

```

0 ( Crew fcns: Engineer                               AWK 10/23/84 )
1 : (/REPAIR) ( -- )
2 (/DAMAGE) 0 NCRS ! 0 .REPAIR-CRS 0 .DAMAGE CTINIT
3 FTRIG OFF FQUIT OFF *SHIP >C+S
4 BEGIN
5 XYSKAN DROP NCRS @ OVER - 7 MOD DUP 0< 7 * + NCRS !
6 IF OCRS @ .REPAIR-CRS NCRS @ DUP .REPAIR-CRS .DAMAGE THEN
7 ?TRIG OCRS @ 2* %ARMOR + >R I 1+ C@ 0> R> %ARMOR = OR AND
8 IF OCRS @ .?DESTROYED NOT
9 IF 64 SCAN-DAMAGE-STATS
10 IF >R I OCRS @ =
11 IF I .REPAIRING ELSE I .?NEW-REPAIR
12 IF I REPAIR> OCRS @ CTCR >REPAIR
13 ELSE I .CONT-REPAIR THEN THEN R> DROP
14 ELSE CTERASE OCRS @ >REPAIR THEN THEN THEN ?QUIT
15 UNTIL OCRS @ .REPAIR-CRS ICLOSE ;

```

14

```

0 ( -----overlay suffix for ENGINEER----- )
1 DISPOSE
2 CLOSE-OVERLAY
3 98 OVERLAY ENGINEER-OV
4 ENGINEER-OV
5 FORTH DEFINITIONS
6
7 : ov/DAMAGE ENGINEER-OV ENGINEER-VOC (/DAMAGE) ;
8 : ov/REPAIR ENGINEER-OV ENGINEER-VOC (/REPAIR) ;
9 : ovDBARS ENGINEER-OV ENGINEER-VOC DBARS ;
10
11 OV-CANCEL
12
13
14
15

```